

# Integrale- Lerndatei

Prof. Dr. Dörte Haftendorn, MuPAD 4, <http://haftendorn.uni-lueneburg.de> Aug.06

Automatische Übersetzung aus MuPAD 3.11, Aug 05 Update Sept.05

*Es fehlen noch textliche Änderungen, die MuPAD 4 direkt berücksichtigen, das ist in Arbeit.*

Web: <http://haftendorn.uni-lueneburg.de> [www.mathematik-verstehen.de](http://www.mathematik-verstehen.de)

+++++

## 1. Grundlagen rechnen und zeichnen ###

## 2. Riemann-Summen ### 3. Zwischenflächen ###

```
f:=x->(x-1)^2-1
```

```
x -> (x - 1)2 - 1
```

### Unbestimmtes Integral

```
int(f(x), x)
```

```
 $\frac{x^3}{3} - x^2$ 
```

### Bestimmtes Integral


```
int(f(x), x=2..4)
```

```
 $\frac{20}{3}$ 
```

Um die vom Integral berechnete gewichtete Fläche zu bestimmen, müssen verschiedenartige Graphik-Elemente kombiniert werden. Sie heißen "Graphik-Primitive", siehe Graphen-Lerndatei, Level 2.

```
gf:=plot::Function2d(f(x), x=-1..4,  
                    Color=RGB::Red, LineWidth=1):
```

```
igf:=plot::Integral(gf):  
plot(igf, gf)
```




**Der Befehl `plot::Integral` bekommt also die eine Graphik-Primitive der Funktion  $f$  übergeben.**

**Für Integrale, deren Intervall nicht das Darstellungsintervall ist, muss man einen**

**kurzen Graphen in dem Integrationsintervall erzeugen.**

```
gfklein:=plot::Function2d(f(x),x=2.5..3.5):  
igfklein:=plot::Integral(gfklein):  
plot(igfklein,gf)
```



## **2. Riemann-Summen #####**



Zur Unterstützung von Unterricht gibt es eine einfache Art, Riemannsche Untersummen und Obersummen darzustellen.


Das wird durch eine Option in dem `plot::Integral` -Befehl erreicht. Die 3 in dem Befehl steht für 3 Streifen.

```
p:=x->1/2*(x-3)^2+1
```

$$x \rightarrow \frac{(x-3)^2}{2} + 1$$

```
gp:=plot::Function2d(p(x),x=0..5,  
                    Color=RGB::Red, LineWidth=1):  
igprl:=plot::Integral(gp,3,IntMethod=RiemannLower,  
                    Color=RGB::LightYellow,ShowInfo =  
                    [IntMethod = "RiemannLower",  
                    Error = "Differenz",Position=[2,4.5]]):  
igpru:=plot::Integral(gp,3,IntMethod=RiemannUpper):  
plot(igpru,gp);plot(igprl,gp);  
plot(igpru,igprl,gp,Scaling=Constrained);
```






### 3. Zwischenflächen #####

```
h:=x->x*(3-x);
```

```
x → x · (3 - x)
```

```
gh:=plot::Function2d(h(x),x=-1..3):
```

```
plot(gf,gh)
```



Erst braucht man wieder Graphik-Primitive der beteiligten Funktionen.  
 Dann muss in vielen Aufgaben erst noch die Schnittstelle berechnet werden:

```
l0:=solve(f(x)=h(x))
```

```
{[x = 0], [x = 5/2]}
```

```
l02:=(l0[2])[1]
```

$$x = \frac{5}{2}$$

```
xs:=subs(x, 1o2)
```


$$\frac{5}{2}$$

Hier wurde die Lösung herausgriffen und getauft, damit sie nicht von Hand übertragen werden muss.

```
zwischen:=plot::Hatch(gf,gh,0..xs)
```

```
plot::Hatch("Function2d((x-1)^2-1, x=-1..4)", "Function2d(-x*(x-3))
```

```
plot(zwischen,gf,gh)
```



**Berechnung der Zwischenfläche**

```
int(h(x)-f(x),x=0..5/2); float(%);
```

$$\frac{125}{24}$$

5.208333333